

ECGR 6118
Computer Project: Wavelets
Student Name: _____

(copyright T. Weldon, 2006)

For this project, you may use mathcad or NetBeans


Project tasks:

For tasks 1-5 below, use the Daubechies 8 filters.


Turn in a 7-page report, with page-1 cover sheet, page 2 explaining the method you used (Daubechies or ideal lowpass), pages 3 - 7 being 6x6 inch printouts of your multiresolution images for tasks 1-5, as illustrated below.

In an appendix, turn in your source code for your wavelet transform (mathcad or java).

1. Form a 2-level multiresolution wavelet transform of the wheel image, as shown below ().



2. Form a 2-level multiresolution wavelet transform of the camera image, as shown below (you may use approximately ideal lowpass/highpass filters, and may "zero-out" the problem frequencies for expediency).



3. Form a 2-level multiresolution wavelet transform of the dslit image.

4. Form a 2-level multiresolution wavelet transform of the nsin image.

5. Form a 2-level multiresolution wavelet transform of the rect17 image.

Daubechies wavelets

Daub4 is the impulse response of the 4-point long Daubechies wavelet,
 Daub8 is the impulse response of the 8-point long wavelet.

You may use either one of them, if you decide to use to use this type of wavelet.

$$\text{daub4} := \begin{pmatrix} 0.48296 \\ 0.8365 \\ 0.22414 \\ -0.12941 \end{pmatrix}$$

$$\text{daub8} := \begin{pmatrix} 0.2303778 \\ 0.7148465 \\ 0.6308808 \\ -0.0279838 \\ -0.1870348 \\ 0.0308414 \\ 0.0328830 \\ -0.0105974 \end{pmatrix}$$


Order of high pass coefficients are the lowpass coefficients in reverse order and multiplied by $(-1)^n$


$$\text{daub4hp} := \begin{pmatrix} -0.12941 \\ -0.22414 \\ 0.8365 \\ -0.48296 \end{pmatrix}$$

calculate the frequency response:

$$\text{Hdaub}(w, \text{coeff}) := \sum_{n=0}^{\text{rows}(\text{coeff})-1} \text{coeff}_n \cdot e^{-i \cdot w \cdot n}$$

$$\text{magHdaub}(w, \text{coeff}) := |\text{Hdaub}(w, \text{coeff})|$$





Applying the wavelet filters

Let us next demonstrate how to apply the wavelet filters.
 First, we will construct a small 8x8 test image (matrix).
 Then we will construct the Daub4 vertical and horizontal impulse responses,
 take their fourier transforms, and construct the 4 possible filters in the
 frequency domain (LoLo,LoHi,HiLo,HiHi)
 Then, apply the filters to the test image.
 Finally, downsample the images and combine into one final image.

Repeat the above process on the LoLo image to get multiple-level wavelet decompositions.


construct a small 8x8 test image

```

N := 8      r := 0..N - 1      c := r
xr,c := 0
rr := 3..4      cc := rr
xrr,cc := 255

```

$$x = \begin{pmatrix}
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 255 & 255 & 0 & 0 & 0 \\
0 & 0 & 0 & 255 & 255 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0
\end{pmatrix}$$



Next create frequency-domain filters

Note: For convolution, it is more convenient to remove the factor of 1/N from the 2D fourier transform `cfft()`, so we multiply by N below, and later divide by N when taking the inverse transforms

```
N := rows(daubv)      N = 8

DvLo := N·cfft(daubv)   DvHi := N·cfft(daubvhp)  vertical filters, lowpass & highpass

DhLo := N·cfft(daubh)   DhHi := N·cfft(daubhhp)  horizontal filters
```


Finally, create the four wavelet frequency-domain filters

```
LoLo := (DvLo·DhLo)
LoHi := (DvLo·DhHi)
HiLo := (DvHi·DhLo)
HiHi := (DvHi·DhHi)
```

Note: you must use the Mathcad vectorize operator (arrow-overstrike) to do element-by-element multiplication instead of the default matrix multiplication

```
magF := 100  $\overrightarrow{|LoLo|}$ 
```

check the LoLo frequencyresponse by plotting it



Finally, filter the image using Fourier

```
X=Fourier[x] where x is the image to be filtered
xLoLo = inverseFourier[ X LoLo ]
xLoHi = inverseFourier[ X LoHi ]
and so forth
```

As mentioned above, for convolution, it is more convenient to remove the factor of 1/N from the 2D fourier transform `cfft()`, so we multiply by N below, and later divide by N when taking the inverse transforms

$$X := \text{Ncfft}(x)$$

$$x\text{LoLo} := \frac{1}{N} \cdot \text{icfft}(\overrightarrow{(X \cdot \text{LoLo})})$$

$$x\text{LoHi} := \frac{1}{N} \cdot \text{icfft}(\overrightarrow{(X \cdot \text{LoHi})})$$

$$x\text{HiLo} := \frac{1}{N} \cdot \text{icfft}(\overrightarrow{(X \cdot \text{HiLo})})$$



$$x\text{HiHi} := \frac{1}{N} \cdot \text{icfft}(\overrightarrow{(X \cdot \text{HiHi})})$$

Check that everything works:

Rescale the xLoLo result to plot it (daubechies LoLo filter gain is 2), and 127 is added to allow plotting of negative values

$$xx := \frac{x\text{LoLo}}{4} + 127$$

note that for this very small image, the result is noticeably shifted by approx 2 pixels because the daubechies impulse response is not "centered around zero"



Note that daubechies LoLo filter gain at DC (0 Hz) is 2 as shown below

$$\text{LoLo}_{0,0} = 2$$

$$\sum_{rr=0}^{\text{rows}(x)-1} \sum_{cc=0}^{\text{rows}(x)-1} x_{rr,cc} = 1.02 \times 10^3$$

$$\sum_{rr=0}^{\text{rows}(x\text{LoLo})-1} \sum_{cc=0}^{\text{rows}(x\text{LoLo})-1} x\text{LoLo}_{rr,cc} = 2.04 \times 10^3$$

Downsample the filtered images and combine into Wavelet image

to save time, write a simple downsampling routine

```
downsample(img) :=  $\left\{ \begin{array}{l} z \leftarrow 0 \\ \text{for } rr \in 0.. \frac{\text{rows}(\text{img})}{2} - 1 \\ \quad \text{for } cc \in 0.. \frac{\text{cols}(\text{img})}{2} - 1 \\ \quad \quad \text{subs}_{rr,cc} \leftarrow \text{img}_{2 \cdot rr, 2 \cdot cc} \\ \text{return subs} \end{array} \right.$ 
```

$x_{LoLoSub} := \text{downsample}(x_{LoLo})$ $x_{LoHiSub} := \text{downsample}(x_{LoHi})$

$x_{HiLoSub} := \text{downsample}(x_{HiLo})$ $x_{HiHiSub} := \text{downsample}(x_{HiHi})$


$$x_{LoLoSub} = \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 443.949 & 31.873 \\ 0 & 0 & 31.873 & 2.288 \end{pmatrix}$$

Finally, combine the images to create the wavelet image

```
xWavelet := stack(augment(xLoLosub, xLoHisub), augment(xHiLosub, xHiHisub))
```

rescale the answer so it can be displayed as an image

```
xWaveletImage := 127 +  $\frac{xWavelet}{4}$ 
```



To do multiple scales, repeat the above process on the xLoLo lowpass image

Finally, you could do everything above as a Mathcad routine

```

wavelet(img, filterlp, filterhp) :=
  z ← 0
  daubv ← img·0
  daubvhp ← img·0
  for rr ∈ 0..rows(filterlp) - 1
    daubvrr,0 ← filterlprr
  for rr ∈ 0..rows(filterhp) - 1
    daubvhprr,0 ← filterhprr
  daubh ← daubvT
  daubhhp ← daubvhpT
  N ← rows(daubv)
  DvLo ← N·cfft(daubv)
  DvHi ← N·cfft(daubvhp)
  DhLo ← N·cfft(daubh)
  DhHi ← N·cfft(daubhhp)
  LoLo ← (DvLo·DhLo)
  LoHi ← (DvLo·DhHi)
  HiLo ← (DvHi·DhLo)
  HiHi ← (DvHi·DhHi)
  X ← Ncfft(img)
  xLoLo ←  $\frac{1}{N} \cdot \text{icfft}(\overrightarrow{(X \cdot \text{LoLo})})$ 
  xLoHi ←  $\frac{1}{N} \cdot \text{icfft}(\overrightarrow{(X \cdot \text{LoHi})})$ 
  xHiLo ←  $\frac{1}{N} \cdot \text{icfft}(\overrightarrow{(X \cdot \text{HiLo})})$ 
  xHiHi ←  $\frac{1}{N} \cdot \text{icfft}(\overrightarrow{(X \cdot \text{HiHi})})$ 
  z0 ← downsample(xLoLo)
  z1 ← downsample(xLoHi)
  z2 ← downsample(xHiLo)
  z3 ← downsample(xHiHi)
  return z

```

create filters
impulse responses

create filters
freq response

create four
wavelet filter
freq responses

Filter the image
using all 4 filters

Return a vector z
containing all 4
downsampled
results

Try out the above function

```
zz := wavelet(x, daub4, daub4hp)
```


as before, combine the images to create the wavelet image

```
zzWavelet := stack(augment(zz0, zz1), augment(zz2, zz3))
```

rescale the answer so it can be displayed as an image

```
zzWaveletImage := 127 +  $\frac{zzWavelet}{4}$ 
```

and you should get the same result below, as previously obtained



Use a function as follows, to read images

```
infile := READBMP("wheel.gif")    x := infile
```